



Emily Riehl

Johns Hopkins University

## The $\infty$ -cosmos project

Formalizing 1-, 2-,  $\mathcal{V}$ -, and  $\infty$ -category theory in Lean



The  $\infty$ -cosmos project aims to leverage the existing libraries developing

- 1-category theory,
- 2-category theory, and
- enriched ( $\mathcal{V}$ -)category theory

in Lean to formalize basic  $\infty$ -category theory.

After giving a high-level overview of the problem, plan, and progress of the project (so far), we illustrate some challenges that we have encountered related to the formalization of supporting results from 1-category theory, 2-category theory, and  $\mathcal{V}$ -category theory in hopes of attracting interest from folks who want to help us solve them.



1. What are  $\infty$ -categories?
2. The problem, the plan, and progress
3. Challenges



1

What are  $\infty$ -categories?

# The idea of an $\infty$ -category



Lean defines an ordinary 1-category as follows:

```
class Quiver (V : Type u) where
  /-- The type of edges/arrows/morphisms between a given source and target. -/
  Hom : V → V → Sort v

class CategoryStruct (obj : Type u) extends Quiver.{v + 1} obj : Type max u (v + 1) where
  /-- The identity morphism on an object. -/
  id : ∀ X : obj, Hom X X
  /-- Composition of morphisms in a category, written `f >> g`. -/
  comp : ∀ {X Y Z : obj}, (X → Y) → (Y → Z) → (X → Z)

class Category (obj : Type u) extends CategoryStruct.{v} obj : Type max u (v + 1) where
  /-- Identity morphisms are left identities for composition. -/
  id_comp : ∀ {X Y : obj} (f : X → Y), 1 X >> f = f := by aesop_cat
  /-- Identity morphisms are right identities for composition. -/
  comp_id : ∀ {X Y : obj} (f : X → Y), f >> 1 Y = f := by aesop_cat
  /-- Composition in a category is associative. -/
  assoc : ∀ {W X Y Z : obj} (f : W → X) (g : X → Y) (h : Y → Z), (f >> g) >> h = f >> g >> h := by
    aesop_cat
```

The idea of an  $\infty$ -category is just to

- replace all the types by  $\infty$ -groupoids aka homotopy types aka anima, i.e., the information of a topological space encoded by its homotopy groups
- and suitably weaken all the structures and axioms.

# Quasi-categories in Lean



An elegant “coordinatization” of these ideas encodes an  $\infty$ -category as a **quasi-category**, which Johan Commelin contributed to Mathlib:

```
/-- A simplicial set `S` is a *quasicategory* if it satisfies the following horn-filling condition:  
for every `n : ℕ` and `0 < i < n`,  
every map of simplicial sets `σ₀ : Δ[n, i] → S` can be extended to a map `σ : Δ[n] → S`.
```

```
[Kerodon, 003A] -/
```

```
class Quasicategory (S : SSet) : Prop where  
  hornFilling' : ∀ {n : ℕ} {i : Fin (n+3)} (σ₀ : Δ[n+2, i] → S)  
    (_h₀ : 0 < i) (_hn : i < Fin.last (n+2)),  
    ∃ σ : Δ[n+2] → S, σ₀ = hornInclusion (n+2) i >> σ
```

Here a simplicial set<sup>1</sup>  $S$  is a **quasi-category** if it satisfies a certain property: namely if any **inner horn**  $\sigma_0$  in  $S$  can be extended to a simplex  $\sigma$ .

$$\begin{array}{ccc} \Delta[n+2, i] & \xrightarrow{\sigma_0} & S \\ \text{hornInclusion } (n+2) \ i \downarrow & \nearrow \sigma & \\ \Delta[n+2] & & \end{array}$$

---

<sup>1</sup>A **simplicial set** is a contravariant functor from a certain SimplexCategory to Type.

## How are quasi-categories $\infty$ -categories?



A similar “coordinatization” of the notion of  $\infty$ -groupoid is as a Kan complex, a simplicial set in which any **outer horn** can be extended to a simplex.

```
/-- A simplicial set `S` is a *Kan complex* if it satisfies the following horn-filling condition:  
for every nonzero `n : ℕ` and `0 ≤ i ≤ n`,  
every map of simplicial sets `σ₀ : Δ[n, i] → S` can be extended to a map `σ : Δ[n] → S`. -/  
class KanComplex (S : SSet.{u}) : Prop where  
  hornFilling : ∀ {n : ℕ} {i : Fin (n + 2)} (σ₀ : Δ[n + 1, i] → S),  
    ∃ σ : Δ[n + 1] → S, σ₀ = hornInclusion (n + 1) i » σ
```

Then

- the maximal sub Kan complex in a quasi-category  $S$  defines the  **$\infty$ -groupoid of objects**,
- a certain pullback of the exponential  $s\text{Hom}(\Delta[1], S)$  defines the  **$\infty$ -groupoid of arrows between two objects**,
- $n$ -ary composition can be shown to be **well-defined up to a contractible  $\infty$ -groupoid** of choices.

None of this has been formalized in `Mathlib`.



2

The problem, the plan, and progress



## The problem and the plan



The objective is to add the **theory** of  $\infty$ -categories to `Mathlib`, not just the **definition**. Textbooks that develop that theory using quasi-categories (Lurie's *Higher Topos Theory*, Cisinski's *Higher Categories and Homotopical Algebra*) tend to be very long.

The idea of the  **$\infty$ -cosmos project** is to develop the theory of  $\infty$ -categories more abstractly, using the axiomatic notion of an  $\infty$ -cosmos, which is an enriched category whose objects are  $\infty$ -categories.

From this we can extract a 2-category whose objects are  $\infty$ -categories, whose morphisms are  $\infty$ -functors, and whose 2-cells are  $\infty$ -natural transformations. The formal theory of  $\infty$ -categories (adjunctions, co/limits, Kan extensions) can be defined using this 2-category and some of these notions are in the `Mathlib` already!

Proving that quasi-categories define an  $\infty$ -cosmos will be hard, but this tedious verifying of homotopy coherences will only need to be done once rather than in every proof.



The  $\infty$ -cosmos project was launched in September 2024. After adding some background material on enriched category theory, we have formalized the following definition:

1.2.1. Definition ( $\infty$ -cosmos). An  $\infty$ -cosmos  $\mathcal{K}$  is a category that is enriched over quasi-categories,<sup>13</sup> meaning in particular that

- its morphisms  $f: A \rightarrow B$  define the vertices of a quasi-category denoted  $\text{Fun}(A, B)$  and referred to as a **functor space**,

that is also equipped with a specified collection of maps that we call **isofibrations** and denote by “ $\twoheadrightarrow$ ” satisfying the following two axioms:

- (i) (completeness) The quasi-categorically enriched category  $\mathcal{K}$  possesses a terminal object, small products, pullbacks of isofibrations, limits of countable towers of isofibrations, and cotensors with simplicial sets, each of these limit notions satisfying a universal property that is enriched over simplicial sets.<sup>14</sup>
- (ii) (isofibrations) The isofibrations contain all isomorphisms and any map whose codomain is the terminal object; are closed under composition, product, pullback, forming inverse limits of towers, and Leibniz cotensors with monomorphisms of simplicial sets; and have the property that if  $f: A \twoheadrightarrow B$  is an isofibration and  $X$  is any object then  $\text{Fun}(X, A) \twoheadrightarrow \text{Fun}(X, B)$  is an isofibration of quasi-categories.

# A formalized definition of an $\infty$ -cosmos



```
variable (K : Type u) [Category.{v} K] [SimplicialCategory K]

/-- A `PreInfinityCosmos` is a simplicially enriched category whose hom-spaces are quasi-categories
and whose morphisms come equipped with a special class of isofibrations.-/
class PreInfinityCosmos extends SimplicialCategory K where
  [has_qcat_homs :  $\forall \{X Y : K\}$ , SSet.Quasicategory (EnrichedCategory.Hom X Y)]
  [IsIsofibration : MorphismProperty K]

variable (K : Type u) [Category.{v} K] [PreInfinityCosmos.{v} K]

/-- An `InfinityCosmos` extends a `PreInfinityCosmos` with limit and isofibration axioms.-/
class InfinityCosmos extends PreInfinityCosmos K where
  comp_isIsofibration {A B C : K} (f : A  $\rightarrow$  B) (g : B  $\rightarrow$  C) : IsIsofibration (f.1  $\gg$  g.1)
  iso_isIsofibration {X Y : K} (e : X  $\rightarrow$  Y) [IsIso e] : IsIsofibration e
  all_objects_fibrant {X Y : K} (hY : IsConicalTerminal Y) (f : X  $\rightarrow$  Y) : IsIsofibration f
  [has_products : HasConicalProducts K]
  prod_map_fibrant { $\gamma$  : Type w} {A B :  $\gamma \rightarrow$  K} (f :  $\forall i$ , A i  $\rightarrow$  B i) :
    IsIsofibration (Limits.Pi.map ( $\lambda i \mapsto$  (f i).1))
  [has_isoFibration_pullbacks {E B A : K} (p : E  $\rightarrow$  B) (f : A  $\rightarrow$  B) : HasConicalPullback p.1 f]
  pullback_is_isoFibration {E B A P : K} (p : E  $\rightarrow$  B) (f : A  $\rightarrow$  B)
    (fst : P  $\rightarrow$  E) (snd : P  $\rightarrow$  A) (h : IsPullback fst snd p.1 f) : IsIsofibration snd
  [has_limits_of_towers (F :  $\mathbb{N}^{\text{op}} \Rightarrow$  K) :
    ( $\forall n : \mathbb{N}$ , IsIsofibration (F.map (homOfLE (Nat.le_succ n)).op))  $\rightarrow$  HasConicalLimit F]
  has_limits_of_towers_isIsofibration (F :  $\mathbb{N}^{\text{op}} \Rightarrow$  K) (hf) :
    haveI := has_limits_of_towers F hf
    IsIsofibration (limit. $\pi$  F (.op 0))
  [has_cotensors : HasCotensors K]
  leibniz_cotensor {U V : SSet} (i : U  $\rightarrow$  V) [Mono i] {A B : K} (f : A  $\rightarrow$  B) {P : K}
    (fst : P  $\rightarrow$  U  $\bowtie$  A) (snd : P  $\rightarrow$  V  $\bowtie$  B)
    (h : IsPullback fst snd (cotensorCovMap U f.1) (cotensorContraMap i B)) :
    IsIsofibration (h.isLimit.lift <|
      PullbackCone.mk (cotensorContraMap i A) (cotensorCovMap V f.1)
        (cotensor_bifunctoriality i f.1)) --TODO : Prove that these pullbacks exist.
  local_isoFibration {X A B : K} (f : A  $\rightarrow$  B) : Isofibration (toFunMap X f.1)
```

## Related contributions to Mathlib



One successful aspect of our project is the rapid rate of contributions to Mathlib:

- codiscrete categories (Alvaro Belmonte)
- reflexive quivers (Mario Carneiro, Pietro Monticone, Emily Riehl)
- the opposite category of an enriched category (Daniel Carranza)
- a closed monoidal category is enriched in itself (Daniel Carranza, Joël Riou)
- StrictSegal simplicial sets are 2-coskeletal (Mario Carneiro and Joël Riou)
- StrictSegal simplicial sets are quasicategories (Johan Commelin, Emily Riehl, Nick Ward)
- left and right lifting properties (Jack McKoen)
- SSet.hoFunctor, which constructs a category from a simplicial set (Mario Carneiro, Pietro Monticone, Emily Riehl, Joël Riou)
- SimplicialSet (co)skeleton properties (Mario Carneiro, Pietro Monticone, Emily Riehl, Joël Riou)

A key challenge is the extraordinary demands this has placed on Joël Riou as a reviewer.



3

## Challenges

## A challenge from 1-category theory



To define the 2-categorical quotient of an  $\infty$ -cosmos (WIP), Mario Carneiro and I introduced **reflexive quivers**

```
/-- A reflexive quiver extends a quiver with a specified arrow `id X : X → X` for each `X` in its
type of objects. We denote these arrows by `id` since categories can be understood as an extension
of refl quivers.
-/
class ReflQuiver (obj : Type u) extends Quiver.{v} obj : Type max u v where
  /-- The identity morphism on an object. -/
  id : ∀ X : obj, Hom X X
```

and formalized the **free category** and **underlying reflexive quiver** adjunction between **Cat** and **ReflQuiv**. This is now in **Mathlib**:

```
/--
The adjunction between forming the free category on a reflexive quiver, and forgetting a category
to a reflexive quiver.
-/
nonrec def adj : Cat.freeRefl.{max u v, u} → ReflQuiv.forget :=
  Adjunction.mkOfUnitCounit {
```

# A challenge from 1-category theory, continued



```
left_triangle := by
  ext V
  apply Cat.FreeRefl.lift_unique'
  simp only [id_obj, Cat.free_obj, comp_obj, Cat.freeRefl_obj_α, NatTrans.comp_app,
    forget_obj, whiskerRight_app, associator_hom_app, whiskerLeft_app, id_comp,
    NatTrans.id_app']
  rw [Cat.id_eq_id, Cat.comp_eq_comp]
  simp only [Cat.freeRefl_obj_α, Functor.comp_id]
  rw [← Functor.assoc, ← Cat.freeRefl_naturality, Functor.assoc]
  dsimp [Cat.freeRefl]
  rw [adj.counit.component_eq' (Cat.FreeRefl V)]
  conv =>
    enter [1, 1, 2]
    apply (Quiv.comp_eq_comp (X := Quiv.of _) (Y := Quiv.of _) (Z := Quiv.of _) ..).symm
  rw [Cat.free.map_comp]
  show (_ » ((Quiv.forget » Cat.free).map (X := Cat.of _) (Y := Cat.of _))
    (Cat.FreeRefl.quotientFunctor V)) » _ = _
  rw [Functor.assoc, ← Cat.comp_eq_comp]
  conv => enter [1, 2]; apply Quiv.adj.counit.naturality
  rw [Cat.comp_eq_comp, ← Functor.assoc, ← Cat.comp_eq_comp]
  conv => enter [1, 1]; apply Quiv.adj.left_triangle_components V.toQuiv
  exact Functor.id_comp _
```

Lean was confused by

- what category we're in when objects are type classes
- competing notations for functors
- whiskered commutative diagrams

## A challenge from enriched category theory



What is an enriched category?

To borrow a distinction used by Peter May, the term “enriched” can be used as a compound noun — *enriched categories* — or as an adjective — *enriched categories*. In the noun form, an enriched category  $\mathcal{C}$  has no preexisting underlying ordinary category, although we shall see ... that the underlying unenriched 1-category can always be identified a posteriori. When used as an adjective, an enriched category  $\mathcal{C}$  is perhaps most naturally an ordinary category, whose hom-sets can be given additional structure.

— *Elements of  $\infty$ -Category Theory*, Appendix A

Mathlib has both notions, referred to as [enriched categories](#) and [enriched ordinary categories](#), respectively.



# A challenge from enriched category theory, continued



```
variable (V : Type v) [Category.{w} V] [MonoidalCategory V]
class EnrichedCategory (C : Type u1) where
  Hom : C → C → V
  id (X : C) : 1V → Hom X X
  comp (X Y Z : C) : Hom X Y ⊗ Hom Y Z → Hom X Z
  id_comp (X Y : C) : (λ_ (Hom X Y)).inv » id X ▷ _ » comp X X Y = 1 _ := by aesop_cat
  comp_id (X Y : C) : (ρ_ (Hom X Y)).inv » _ ◁ id Y » comp X Y Y = 1 _ := by aesop_cat
  assoc (W X Y Z : C) : (α_ _ _).inv » comp W X Y ▷ _ » comp W Y Z =
    _ ◁ comp X Y Z » comp W X Z := by aesop_cat
```

```
variable (V : Type u') [Category.{v'} V] [MonoidalCategory V]
(C : Type u) [Category.{v} C]
```

```
/-- An enriched ordinary category is a category `C` that is also enriched
over a category `V` in such a way that morphisms `X → Y` in `C` identify
to morphisms `1V → (X → [V] Y)` in `V`. -/
```

```
class EnrichedOrdinaryCategory extends EnrichedCategory V C where
  /-- morphisms `X → Y` in the category identify morphisms
  `1V → (X → [V] Y)` in `V` -/
  homEquiv {X Y : C} : (X → Y) ≈ (1V → (X → [V] Y))
  homEquiv_id (X : C) : homEquiv (1 X) = eId V X := by aesop_cat
  homEquiv_comp {X Y Z : C} (f : X → Y) (g : Y → Z) :
    homEquiv (f » g) = (λ_ _).inv » (homEquiv f ⊗ homEquiv g) »
      eComp V X Y Z := by aesop_cat
```

The enriched categories literature is less clear about this distinction.

## A challenge from 2-category theory



On paper, 2-cells in a 2-category compose by pasting:

$$\begin{array}{ccccccc}
 & & A & \xrightarrow{G_1} & C & \xlongequal{\quad} & C & \xrightarrow{G_2} & E & \xlongequal{\quad} & E \\
 & \nearrow R_1 & \downarrow L_1 & \swarrow \alpha & \downarrow L_2 & \nearrow R_2 & \downarrow L_2 & \swarrow \beta & \downarrow L_3 & \nearrow R_3 \\
 B & \xlongequal{\quad} & B & \xrightarrow{H_1} & D & \xlongequal{\quad} & D & \xrightarrow{H_2} & F & & 
 \end{array}$$

In `Mathlib`, the 2-cells displayed here belong to dependent types (over their boundary 1-cells and objects) and depending on how the whiskerings are encoded are not obviously composable at all:

e.g., is  $R_3 H_2 L_2 \eta_2 G_1 R_1$  composable with  $R_3 H_2 \epsilon_2 L_2 G_1 R_1$ ?

# A challenge from 2-category theory



```
/-- The mates equivalence commutes with vertical composition. -/
theorem mateEquiv_vcomp
  (α : G1 » L2 → L1 » H1) (β : G2 » L3 → L2 » H2) :
  (mateEquiv (G := G1 » G2) (H := H1 » H2) adj1 adj3) (leftAdjointSquare.vcomp α β) =
    rightAdjointSquare.vcomp (mateEquiv adj1 adj2 α) (mateEquiv adj2 adj3 β) := by
  unfold leftAdjointSquare.vcomp rightAdjointSquare.vcomp mateEquiv
  ext b
  simp only [comp_obj, Equiv.coe_fn_mk, whiskerLeft_comp, whiskerLeft_twice, whiskerRight_comp,
    assoc, comp_app, whiskerLeft_app, whiskerRight_app, id_obj, Functor.comp_map,
    whiskerRight_twice]
  slice_rhs 1 4 => rw [- assoc, ← assoc, ← unit_naturality (adj3)]
  rw [L3.map_comp, R3.map_comp]
  slice_rhs 2 4 =>
    rw [- R3.map_comp, ← R3.map_comp, ← assoc, ← L3.map_comp, ← G2.map_comp, ← G2.map_comp]
    rw [- Functor.comp_map G2 L3, β.naturality]
  rw [(L2 » H2).map_comp, R3.map_comp, R3.map_comp]
  slice_rhs 4 5 =>
    rw [- R3.map_comp, Functor.comp_map L2 _, ← Functor.comp_map _ L2, ← H2.map_comp]
    rw [adj2.counit.naturality]
  simp only [comp_obj, Functor.comp_map, map_comp, id_obj, Functor.id_map, assoc]
  slice_rhs 4 5 =>
    rw [- R3.map_comp, ← H2.map_comp, ← Functor.comp_map _ L2, adj2.counit.naturality]
  simp only [comp_obj, id_obj, Functor.id_map, map_comp, assoc]
  slice_rhs 3 4 =>
    rw [- R3.map_comp, ← H2.map_comp, left_triangle_components]
  simp only [map_id, id_comp]
```

In the 2-category [Cat](#), I formalized a proof that the unit  $\eta_2$  and counit  $\epsilon_2$  cancel, but not via a 2-categorical pasting argument. As a result, [Mathlib](#) does not know this is true in any 2-category.

## Contributors to the $\infty$ -cosmos project



Pietro Monticone and Patrick Massot helped us set up a blueprint (and website) to organize the workflow. So far formalizations (and preliminary mathematical work) have been contributed by:

Dagur Asgeirsson, Alvaro Belmonte, Mario Carneiro, Daniel Carranza, Johan Commelin, Jack McKoen, Pietro Monticone, Matej Penciak, Nima Rasekh, Emily Riehl, Joël Riou, Joseph Tooby-Smith, Adam Topaz, Dominic Verity, Nick Ward, and Zeyi Zhao.

Anyone is welcome to join us!

[emilyriehl.github.io/infinity-cosmos](https://emilyriehl.github.io/infinity-cosmos)

Thank you!